

#### Tema 4:

# Especificación de sistemas digitales usando VHDL

Diseño automático de sistemas

#### José Manuel Mendías Cuadros

Dpto. Arquitectura de Computadores y Automática Universidad Complutense de Madrid



# rema 4.

#### DAS

## Contenidos



- ✓ VHDL "sintetizable"
- Especificación usando VHDL.
- ✓ Tipos de datos y operadores.
- ✓ Paquetes estándar.
- ✓ Paquetes no estándar.
- ✓ Lógica combinacional.
- Ejemplos de lógica combinacional.
- ✓ Lógica secuencial.
- ✓ Ejemplos de lógica secuencial.
- ✓ Mezclando VHDL.

## VHDL "sintetizable"



- VHDL es un lenguaje de modelado, no de especificación
  - Todo el HW existente puede ser modelado, pero no todo lo que se especifica puede ser sintetizado.
  - o Desde la óptica de las herramientas EDA, las construcciones VHDL se clasifican en:
    - Soportadas: especifican claramente una funcionalidad HW que debe sintetizarse.
    - Ignoradas: pueden encontrarse en el código fuente VHDL, pero no se tienen en cuenta durante el proceso de síntesis.
    - Prohibidas: no pueden aparecer en el código fuente VHDL, si lo hacen se aborta el proceso de síntesis.
  - El subconjunto sintetizable (construcciones soportadas) y su semántica puede variar de herramienta a herramienta.
    - La semántica de simulación es estándar y la respetan todos los simuladores
    - La semántica de síntesis aunque estandarizada, no se respeta.
- VHDL es un lenguaje orientado a HW que permite escribir SW
  - o Es necesario cambiar de mentalidad, pensar en software solo trae problemas.
- El uso de VHDL y herramientas EDA, no evita diseñar
  - o Antes de codificar, hay que diseñar el sistema hasta un cierto nivel de abstracción.

## VHDL "sintetizable"

#### familia de estándares



- El lenguaje VHDL está estandarizado por el IEEE.
- Desde la óptica de las herramientas EDA son interesantes:
  - 1076 IEEE Standard VHDL Language Reference Manual
    - Define la sintáxis completa del lenguaje y su semántica de simulación.
  - 1164 IEEE Standard Multivalue Logic System for VHDL Model Interoperability
    - Define tipos (atómicos y vectoriales) multivaluados para representar señales digitales.
  - 1076.3 IEEE Standard VHDL Synthesis Packages
    - Define tipos numéricos enteros con y sin signo como vectores de tipo multivaluado.
  - o 1076.6 IEEE Standard for VHDL Register Transfer Level (RTL) Synthesis
    - Define un subconjunto sintetizable del lenguaje y su semántica de síntesis.
- Otros estándares, no directamente relacionados con síntesis son:
  - 1076.1 IEEE Standard VHDL Analog and Mixed-Signal Extensions
  - 1076.1.1 IEEE Standard for VHDL Analog and Mixed-Signal Extensions -- Packages for Multiple Energy Domain Support
  - 1076.2 IEEE Standard VHDL Mathematical Packages
  - 1076.4 IEEE Standard VITAL ASIC (Application-Specific Integrated Circuit) Modeling Specification

# Especificación usando VHDL

- Pensar siempre en HW síncrono de nivel RT (las herramientas que utilizaremos comienzan desde este nivel de abstracción).
  - No comenzar a codificar hasta tener el diagrama de bloques RTL (módulos combinacionales y registros).
    - El uso de VHDL no evita el diseño manual de la estructura RT, lo que evita es el diseño lógico de los bloques combinacionales/secuenciales y la proyección tecnológica.
  - Cada bloque RT habitualmente deberá codificarse con una sentencia concurrente.
    - Se permiten especificaciones más abstractas, pero para que puedan obtenerse soluciones válidas es necesaria una codificación muy cuidadosa.
  - Nunca mezclar en un mismo diseño diferentes estilos de temporización:
    - En particular, usaremos temporización por flanco de subida de un único reloj
  - o El núcleo fundamental de la lógica a sintetizar se especificará mediante expresiones.
    - El resto de construcciones del lenguaje se utilizarán para estructurar el diseño y para especificar su temporización.
- En muchos sentidos VHDL permite "capturar esquemas" textualmente.
  - Ya que la estructura RTL especificada se respeta durante la síntesis.

#### ىنە ئى DAS

## Tipos de datos



- Las herramientas EDA soportan los tipos predefinidos siguientes:
  - o Boolean, bit y bit\_vector
  - O Character y string
  - Integer, positive y natural
    - No se soportan con precisión infinita, el rango deberá poder inferirse del código o especificarse de forma explícita.
    - Todo objeto se implementa mediante el vector de bits más pequeño capaz de representar el rango especificado.
    - Si este rango incluye números negativos, la representación usada será complemento a 2, en otro caso binaria pura.
- También soportan los tipos siguientes:
  - o std\_ulogic, std\_ulogic\_vector, std\_logic, std\_logic\_vector
    - Definidos en el paquete estándar ieee.std\_logic\_1164
  - o signed, unsigned
    - Definidos en los paquetes ieee.numeric\_bit / ieee.numeric\_std / std\_logic\_arith
    - Representan enteros como vectores de bits codificados en complemento a 2 o en binario puro.

## Tipos de datos



Si un tipo está soportado, también lo están todos los operadores asociados a ellos:

- O Lógicos: and | or | nand | nor | xor | not
- o Relacionales: = | /= | < | <= | > | >=
- o Aritméticos: + | | \* | / | mod | rem | abs
- o Concatenación: &
- La definición de tipos por parte del diseñador también se soporta:
  - o Enumerados
    - Todo objeto se implementa mediante el vector de bits más pequeño capaz de representar todos los valores posibles del tipo.
    - Por defecto, cada uno de los literales del tipo se representa mediante la codificación en binario puro de su posición dentro de la declaración (aplica a boolean, bit y character).
  - o Arrays
    - Se soportan vectores monodimensionales con índice entero de cualquier tipo soportado.
  - Registros
    - Se soportan registros cuyas componentes sean de cualquier tipo soportado.

## Paquetes estándar

ieee.std\_logic\_1164 (i)

El paquete std\_logic\_1164 define 2 tipos atómicos multivaluados (uno resuelto y otro no) y 2 tipos vectoriales asociados:

```
type std_ulogic is ('U', 'X', '0', '1', 'Z', 'W', 'L', 'H', '-');
type std_ulogic_vector is array ( natural range <> ) of std_ulogic;
subtype std_logic is resolved std_ulogic;
type std_logic_vector is array ( natural range <> ) of std_logic;
```

- También sobrecarga los operadores lógicos (atómicos y vectoriales)
   para que trabajar sobre los tipos definidos:
  - o and, nand, or, nor, xor, xnor, not
  - Este paquete NO define ningún operador aritmético ni relacional.
- Los tipos resueltos (std\_logic y std\_logic\_vector) son el estándar para la especificación del interconexionado entre modelos VHDL
  - Los puertos de un diseño deberán ser siempre de estos tipos.
- Además deben usarse para especificar señales no numéricas.

'U': no inicializado

# Paquetes estándar

ieee.std\_logic\_1164 (ii)



El significado de cada uno de los 9 valores es:

```
    '0' y '1': valores lógicos fuertes
    'L' y 'H': valores lógicos débiles asimilables a '0' y '1'
    'Z': alta impedancia
    '-': indiferencia (don't care)
    'X': valor fuerte desconocido
    'W': valor débil desconocido
```

- La comparación con valores metalógicos no tiene sentido en síntesis:
  - o Una comparación de igualdad con un valor metalógico es siempre falsa.
  - o Una comparación de desigualdad con un valor metalógico es siempre cierta.

#### یں SAD

## Paquetes estándar

ieee.numeric\_std (i)

- El paquete numeric\_std define 2 tipos de datos vectoriales que, basados en std\_logic, son interpretables como valores numéricos codificados.
  - La longitud del vector determina el rango de valores representables.
  - El tipo unsigned especifica un número natural representado en binario puro.

```
type unsigned is array (natural range <>) of std_logic;
```

O El tipo signed especifica un número entero representado en complemento a 2

```
type signed is array (natural range <>) of std_logic;
```

- Estos tipos son compatibles con std\_logic\_vector por casting.
- También define:
  - Funciones de conversión entre los tipos: integer, natural, unsigned y signed
  - Operadores sobrecargados aritméticos y relacionales con argumentos mixtos de los tipos natural/unsigned o integer/signed con resultados unsigned o signed
  - o Operadores sobrecargados lógicos para los tipos: unsigned y signed
  - Funciones de redimensionado, desplazamiento y rotación para los tipos: unsigned
     y signed

## Paquetes estándar

ieee.numeric\_std (ii)



#### Perfiles de las funciones aritméticas

```
function "+"(L,R: UNSIGNED) return UNSIGNED;
function "+"(L,R: SIGNED) return SIGNED;
function "+"(L: UNSIGNED; R: NATURAL) return UNSIGNED;
function "+"(L: NATURAL; R: UNSIGNED) return UNSIGNED;
function "+"(L: INTEGER; R: SIGNED) return SIGNED;
function "+"(L: SIGNED; R: INTEGER) return SIGNED;
```

#### Perfiles de las funciones relacionales

```
function ">"(L,R: UNSIGNED) return BOOLEAN;
function ">"(L,R: SIGNED) return BOOLEAN;
function ">"(L: NATURAL; R: UNSIGNED) return BOOLEAN;
function ">"(L: INTEGER; R: SIGNED) return BOOLEAN;
function ">"(L: UNSIGNED; R: NATURAL) return BOOLEAN;
function ">"(L: SIGNED; R: INTEGER) return BOOLEAN;
```

#### Perfiles de las funciones de conversión


```
function TO_INTEGER(ARG: UNSIGNED) return NATURAL;
function TO_INTEGER(ARG: SIGNED) return INTEGER;
function TO_UNSIGNED(ARG,SIZE: NATURAL) return UNSIGNED;
function TO_SIGNED(ARG: INTEGER; SIZE: NATURAL) return SIGNED;
function TO_UNSIGNED(ARG: STD_LOGIC_VECTOR) return UNSIGNED;
function TO_SIGNED(ARG: STD_LOGIC_VECTOR) return SIGNED;
function TO_STDLOGICVECTOR(ARG: UNSIGNED) return STD_LOGIC_VECTOR;
function TO_STDLOGICVECTOR(ARG: SIGNED) return STD_LOGIC_VECTOR;
```

## Paquetes estándar

ieee.numeric\_std (iii)



- Los tipos numéricos se usan internamente en los modelos VHDL
  - Para pasar de un tipo a otro se usa casting o funciones de conversión.
  - El uso de casting o funciones de conversión no implica mayor coste hardware.



## Paquetes estándar

ieee.numeric\_std (iv)



- Cuando ambos argumentos son vectoriales, la anchura del resultado es la anchura del máximo de los argumentos
- Cuando solo un argumento es vectorial, la anchura del resultado es la anchura de dicho argumento.
- En los operadores relacionales, el uso de argumentos de distintos tipos y distinta anchura puede dar lugar a paradojas:

arg1	ор	arg2	unsigned	signed	std_logic_vector
"000"	=	"000"	true	true	true
"00"	=	"000"	true	true	false
"100"	=	"0100"	true	false	false
"000"	<	"000"	false	false	false
"00"	<	"000"	false	false	true
"100"	<	"0100"	false	true	false

## Paquetes estándar

ieee.numeric\_std (v)


- Aunque los tipos signed/unsigned sean vectores de std\_logic, no es necesario que las constantes numéricas sean binarias:
  - Todas las funciones aritméticas y relacionales permiten que uno de sus argumentos sea entero/natural.
  - Para asignar valores explícitos pueden usarse las funciones de conversión.

```
architecture ...;
    signal a, b, c, d : unsigned(... downto 0);
begin
    ...
    a <= b + 1;
    c <= d when a<100 else to_unsigned(100,c'length);
end;</pre>
```

- Además, VHDL ofrece facilidades para expresar valores explícitos:
  - Las cadenas de bit/std\_logic pueden expresarse (con y sin separadores) en:
    - Binario: "10100110" / "1010\_0110"
    - Hexadecimal: X"10a6" / X"fad2\_0023"
  - o Los enteros pueden expresarse (con y sin separadores) en:
    - Decimal: 4563 / 4\_563
    - Binario: 2#10100110# / 2#1010 0110#
    - Hexadecimal 16#10a6# / 16#fad2 0023#

# Paquetes no estándar

std\_logic\_unsigned / std\_logic\_signed / std\_logic\_arith (i)



- El paquete std\_logic\_unsigned define operadores aritméticos y relacionales con argumentos mixtos de tipos: integer, natural y std\_logic\_vector
  - Interpreta directamente los argumentos de tipo **std\_logic\_vector** como números sin signo representados en binario puro.
  - Se usaban internamente en módulos que manipulen solo datos sin signo.
- o El paquete std\_logic\_signed define operadores aritméticos y relacionales con
  argumentos mixtos de tipos: integer, natural y std\_logic\_vector
  - Interpreta directamente los argumentos de tipo **std\_logic\_vector** como números con signo representados en complemento a 2.
  - Se usaban internamente en módulos que manipulen solo datos con signo.
- El paquete std\_logic\_arith define
  - 2 tipos vectoriales de std\_logic: unsigned y signed
  - Operadores aritméticos y relaciones con argumentos mixtos de los tipos integer, natural, std\_ulogic, unsigned o signed y con resultados std\_logic\_vector, unsigned o signed
  - A diferencia de numeric\_std, este paquete define operadores con todas las combinaciones posibles de tipo de argumento y el resultado puede ser std\_logic\_vector.

## Paquetes no estándar

std\_logic\_unsigned / std\_logic\_signed / std\_logic\_arith (ii)

#### Perfiles de las funciones aritméticas

```
function "+"(L: UNSIGNED; R: UNSIGNED) return UNSIGNED;
function "+"(L: SIGNED; R: SIGNED) return SIGNED;
function "+"(L: UNSIGNED; R: SIGNED) return SIGNED;
function "+"(L: SIGNED; R: UNSIGNED) return SIGNED;
function "+"(L: UNSIGNED; R: INTEGER) return UNSIGNED;
function "+"(L: INTEGER; R: UNSIGNED) return UNSIGNED;
function "+"(L: SIGNED; R: INTEGER) return SIGNED;
function "+"(L: INTEGER; R: SIGNED) return SIGNED;
function "+"(L: UNSIGNED; R: STD_ULOGIC) return UNSIGNED;
function "+"(L: STD ULOGIC; R: UNSIGNED) return UNSIGNED;
function "+"(L: SIGNED; R: STD_ULOGIC) return SIGNED;
function "+"(L: STD ULOGIC; R: SIGNED) return SIGNED;
function "+"(L: UNSIGNED; R: UNSIGNED) return STD LOGIC VECTOR;
function "+"(L: SIGNED; R: SIGNED) return STD LOGIC VECTOR;
function "+"(L: UNSIGNED; R: SIGNED) return STD LOGIC VECTOR;
function "+"(L: SIGNED; R: UNSIGNED) return STD LOGIC VECTOR;
function "+"(L: UNSIGNED; R: INTEGER) return STD_LOGIC_VECTOR;
function "+"(L: INTEGER; R: UNSIGNED) return STD LOGIC VECTOR;
function "+"(L: SIGNED; R: INTEGER) return STD LOGIC VECTOR;
function "+"(L: INTEGER; R: SIGNED) return STD LOGIC VECTOR;
function "+"(L: UNSIGNED; R: STD_ULOGIC) return STD_LOGIC_VECTOR;
function "+"(L: STD ULOGIC; R: UNSIGNED) return STD LOGIC VECTOR;
function "+"(L: SIGNED; R: STD_ULOGIC) return STD_LOGIC_VECTOR;
function "+"(L: STD ULOGIC; R: SIGNED) return STD LOGIC VECTOR;
```

## Paquetes standard vs. no

El uso de un paquete u otro no afecta a la implementación solo afecta a lo compacta que es la especificación:

```
library ieee;
use ieee.std_logic_1164.all;
entity adder is
port(
    x : in std_logic_vector(7 downto 0);
    y : in std_logic_vector(7 downto 0);
    s : out std_logic_vector(7 downto 0) );
end adder;
library ieee;
use ieee.numeric_std.all;
architecture synl of adder is
begin
    s <= std_logic_vector(unsigned(x) + unsigned(y));
end synl;</pre>
```

```
library ieee;
use ieee.std_logic_unsigned.all;
architecture syn2 of adder is
begin
   s <= x + y;
end syn2;</pre>
```


```
library ieee;
use ieee.std_logic_arith.all;
architecture syn3 of adder is
begin
   s <= unsigned(x) + unsigned(y);
end syn3;</pre>
```




- La lógica combinacional puede ser especificada mediante:
  - Asignaciones concurrentes a señal.
    - En el caso de que sean asignaciones condicionales deben tener rama else.
  - Procesos sin sentencias wait tales que:
    - Todas las señales o variables escritas dentro del proceso sean asignadas al menos una vez en toda activación del mismo.
    - No contengan bucles while o loop y el rango del índice de los bucles for sea estático.
    - Aunque la lista de sensibilidad se ignora, para asegurar la coherencia entre simulación y síntesis, esta debe ser completa (formada por todas las señales leídas dentro del proceso).
- En cualquiera de los casos, cuando se utilizan señales:
  - Los valores iniciales se ignoran.
  - O La definición explícita de retardos (after) se ignora.
  - La asignación de múltiples elementos de forma de onda está prohibida.
- Para asegurar que la lógica especificada es combinacional:
  - Ninguna señal puede formar parte de la/s expresión/es que la definen.

### asignaciones concurrentes (i)

- Toda asignación de señal se implementa como un bloque de lógica combinacional:
  - Con un único puerto de salida (que puede ser vectorial en modelos de nivel RT).
  - O Con tantos puertos de entrada como señales diferentes aparezcan en la expresión.
  - O Con una funcionalidad especificada por los operadores que forman la expresión.



Una posible implementación de la sentencia ya que la implementación definitiva siempre la decide la herramienta




## Lógica combinacional

## asignaciones concurrentes (ii)


- Toda asignación condicional de señal completa (con rama else) se implementa como un bloque de lógica combinacional:
  - Con un único puerto de salida (que puede ser vectorial en modelos de nivel RT).
  - Con tantos puertos de entrada como señales diferentes aparezcan en el lado derecho de la asignación (independientemente de la expresión en la que ocurran).
  - Con un comportamiento que se corresponde con el de un multiplexor 2 a 1 cuyas 3 entradas están conectadas a las salidas de 3 bloques combinacionales.
    - La funcionalidad de dichos bloques queda especificada por los operadores que forman cada una de las 3 expresiones de la sentencia.

señal <= expresión when expresión booleana else expresión;




#### asignaciones concurrentes (iii)

La asignación condicional del valor explícito 'Z' (uno de los valores del tipo std\_logic) especifica la capacidad tri-estado del puerto de salida de la lógica combinacional especificada por el resto de la sentencia.



asignaciones concurrentes (iv)

 Una colección de asignaciones condicionales anidadas se implementa como lógica combinacional en cascada que establece prioridades explícitas entre varios cálculos.





DAS

## Lógica combinacional

#### asignaciones concurrentes (v)


- Toda asignación selectiva de señal se implementa como un bloque de lógica combinacional:
  - Con un único puerto de salida (que puede ser vectorial en modelos de nivel RT).
  - Con tantos puertos de entrada como señales diferentes aparezcan en el lado derecho de la asignación (independientemente de la expresión en la que ocurran).
  - Con un comportamiento que se corresponde con el de un multiplexor 2<sup>n</sup> a 1 cuyas 2<sup>n</sup> entradas están conectadas a las salidas de n+1 bloques combinacionales.
    - Donde n el número de bits con los que se codifica el resultado de la expresión seleccionada.
    - La funcionalidad de dichos bloques queda especificada por los operadores que forman cada una de las n+1 expresiones de la sentencia.





```
with c select
  d <= (a or b) when "00",
      (a and b) when "11" | "10",
      '1' when others;</pre>
```

#### procesos equivalentes



#### Asignación de señal

```
c <= a and b;
process (a, b)
begin
  c <= a and b;
end process;</pre>
```

#### Asignación condicional de señal

```
c <= (a and b) when d='1'
    else (a or b);

process (a, b, d)
begin
    if d='1' then
       c <= a and b;
    else
       c <= a or b;
    end if;
end process;</pre>
```


#### Asignación selectiva de señal

```
with a+b select
   c <= d when "0000",
        not d when "1111",
        '1' when others;

process (a, b, d)
begin
   case a+b is
   when "0000" => c <= d;
   when "1111" => c <= not d;
   when others => c <= '1';
   end case;
end process;</pre>
```

## Lógica combinacional

## expresiones (i)



- Las expresiones VHDL, desde el punto de vista de síntesis, pueden clasificarse en:
  - Expresiones computables: aquellas cuyo resultado puede ser determinado en tiempo de análisis, por tanto tienen un valor estático.
    - No requieren HW que las calcule, se implementan como conexiones a VCC y/o GND.
    - Pueden usarse a discreción para facilitar la legibilidad del código.
  - Expresiones no computables: aquellas cuyo resultado no puede ser determinado en tiempo de análisis, por tanto tienen un valor dinámico.
    - Se implementan mediante HW combinacional que las calcula.
- Son expresiones no computables aquellas que contengan:
  - o Puertos.
  - Variables o señales que hayan sido asignadas por una expresión no computable.
  - Variables o señales que hayan sido asignadas (aunque sea por expresiones computables) en función del valor de una condición no computable.



## tem

#### DAS

## Lógica combinacional

#### expresiones (ii)

•••••

computable, v1 vale '1'

computable, v0 vale '1

computable, v2 vale '0'

computable, vint vale 3

computable, vVector vale "1000"

computable, v1 vale '1'

computable, v2 vale '0'

computable, v0 vale '0'

computable, v1 vale '1'

computable, v1 vale '1'

no computable, v2 depende de s

no computable, v1 depende de s


no computable, v2 depende de s


```
architecture ... of ... is
  signal s : std logic;
  function mux(a,b,c:std logic) return std logic is
  begin
    if (c = '1') then return a; else return b; end if;
  end:
  procedure comp( a : std_logic; b : out std_logic ) is
  begin
   b := not a;
  end:
begin
  process (s)
    variable v0, v1, v2 : std logic;
    variable vInt : integer;
    subtype miVector is std logic vector(0 to 3);
    variable vVector : miVector;
  begin
   v0 := '1';
    v1 := v0;
    v2 := not v1;
    for i in 0 to 3 loop vInt := i; end loop;
    vVector := miVector'(v1, v2, '0', '0');
    v1 := mux(v0, v1, v2);
    comp( v1, v2 );
    v0 := s and '0';
    v1 := mux( s, '1', '0' );
    v1 := mux( '1', '1', s );
      (s = '1') then v2 := '0'; else v2 := 1; end if;
    v2 := v1 or '0';
  end process;
end ...;
```

## Lógica combinacional

#### expresiones (iii)

- Las expresiones simples (formadas un único operador) especifican una conducta primitiva sin estructura interna aparente.
- Las expresiones compuestas (formada por varios operadores) o las expresiones en cascada especifican:
  - La propia conducta a diseñar.
  - Una ordenación de los cálculos que se traduce en una estructura inicial sobre la que la herramienta comenzará a optimizar.
  - La estructura inicial es determinante (sobre todo a nivel RT) ya que puede desde facilitar la búsqueda del diseño óptimo hasta ocultarlo.








## Lógica combinacional

#### procesos (i)

- Un proceso se implementará como HW combinacional si y solo si
  - No tiene sentencias wait y todas las señales o variables escritas dentro del mismo se asignan al menos una vez bajo cualquier condición de ejecución.

```
process (a, b, c, d)
begin
  if a='1' then
    e <= c;
  elsif b='1' then
    e <= d;
  else
    e <= '0';
  end if;
end process;</pre>
```


а	b	е
0	0	0
0	1	d
1	0	С
1	1	С



 Si alguna señal o variable no se asigna bajo alguna condición de ejecución, se implementara HW secuencial para ella.

```
process (a, b, c, d)
begin
  if a='1' then
    e <= c;
elsif b='1' then
    e <= d;
end if;
end process;</pre>
```

а	b	е
0	0	no cambia
0	1	d
1	0	С
1	1	С



## Lógica combinacional


### procesos (ii)



Aunque es posible (solo si el tipo de la señal es resuelto) no es recomendable asignar concurrentemente varias veces una misma señal.

```
architecture ...;
  signal d : ...;
begin
  ...
  d <= a or b;
  d <= a and b;
  ...
end ...;</pre>
```


```
architecture ...;
signal d : ...;
begin
...
d <= a and b;
d <= a or b;
...
end ...;</pre>
```




el valor depende de la función de resolución (en simulación) y de la tecnología (en síntesis)

- El orden de asignación secuencial de señales sí importa:
  - Si bajo unas condiciones de ejecución un proceso asigna varias veces una misma señal sólo se diseñará lógica para implementar la última asignación.

```
process (a, b)
begin
  d <= a and b;
  d <= a or b;
end process;</pre>
```







procesos (iii)



En procesos implementados como lógica combinacional, el orden de asignación secuencial de variables no importa.

```
process ( input )
                                                         process ( input )
                          input
  variable a : ...;
                                                           variable a : ...;
begin
                                                         begin
                                              output
  a := input + 1;
                                                           output <= a;
                                                           a := input + 1;
  output <= a;
end process;
                                                         end:
```

Al ser ambos procesos 'combinacionales', actualizan variables y señales en cualquiera de sus ejecuciones, como en el modelo VHDL las variables conservan valores entre llamadas la implementación debe ser en cualquier caso un incrementador.

Esto provoca inconsistencias entre simulación y síntesis y la herramienta avisa de que se lee la variable antes de escribirla.


#### procesos (iv)

- En procesos implementados como lógica secuencial, el orden de asignación secuencial de variables sí importa.
  - Dicho orden determina el número de elementos de memoria que se implementan

```
process (gate, input)
  variable a, b : ...;
begin
  if gate='1' then
    a := input;
    b := a;
    output <= b;
  end if;
end process;</pre>
```

```
input D Q output gate G
```

```
process (gate, input)
  variable a, b : ...;
begin
  if gate='1' then
    b := a;
    a := input;
    output <= b;
  end if;
end process;</pre>
```




- Por ello, se recomienda usar siempre señales para almacenar el estado de un proceso 'secuencial'
  - Las variables se usan para simplificar expresiones en procesos 'combinacionales'

### procesos (v)

- Un error típico al implementar procesos es que se genere lógica secuencial cuando el diseñador quiso especificar lógica combinacional
  - Porque dejó alguna sentencias if sin rama else
  - Porque dejó algún caso sin cubrir en alguna sentencia case (o sin others).
- Para evitar este error se recomienda asignar valores por defecto a todas las variables o señales al comienzo del proceso.
  - Estos valores por defecto serán sobreescritos por el resto del código.

```
process (a, b, c, d)
begin
  e <= '0';
  if a='1' then
        b='1' then
    e \le d;
end process:
```

а	b	е
0	0	0
0	1	d
1	0	С
1	1	С





## Lógica combinacional

reuso (i)

- Habitualmente, cada ocurrencia de un operador da lugar a un bloque de lógica (recurso) que la implementa.
  - Cuando un único bloque de lógica implementa varias ocurrencias de un operador se dice que dichas ocurrencias comparten el recurso.
- Para que dos operaciones compartan un recurso debe cumplirse que:
  - Ocurran dentro del mismo proceso.
  - Sean mutuamente exclusivos: no exista ninguna posible ejecución en la que se necesite calcular simultáneamente el resultado de ambos.

Exista un recurso capaz de ejecutar ambas.





end if;

end if;
end process;

else

if not c1 then
z4 <= k + 1;</pre>

 $z4 \ll m + n;$ 

## Lógica combinacional

### reuso (ii)

- La detección de pares de operaciones mutuamente exclusivas depende de la pericia de la herramienta EDA
  - Típicamente, detectan solo la exclusividad mutua estructural.

```
process (a,b,c,d,e,f,g,h,i,j,k,l,m,n,c1,c2)
begin

z1 <= a + b;
if c1 then
    z2 <= c + d;
else
    z2 <= e + f;
if c2 then
    z3 <= g + h;
else
    z3 <= i + j;
end if;</pre>
mutex a + b
c + d
no
```

Típicamente las herramientas consideran que las condiciones son independientes

mutex	a + b	c + d	e + f	g + h	i + j	k + l	m + n
a + b		no	no	no	no	no	no
c + d	no		si	si	si	no	no
e + f	no	si		no	no	no	no
g + h	no	si	no		si	no	no
i+j	no	si	no	si		no	no
k + l	no	no	no	no	no		si
m + n	no	no	no	no	no	si	

## Lógica combinacional

#### funciones


- Una función (independientemente de su codificación) se implementa siempre como un bloque de lógica combinacional, por ello:
  - Un mismo código puede sintetizarse de manera distinta según se defina o no como función.

```
function foo( d, gate : bit ) return bit is
  variable q: bit;
begin
  if gate='1' then
    q := d;
  end if;
  return q;
end;
...
q <= foo( d, gate );
...</pre>
```

d	gate	q
0	0	0
0	1	0
1	0	0
1	1	1



process ( gate, d )
begin
 if gate='1' then
 q <= d;
 end if;
end process;</pre>



Las variables de funciones son dinámicas, por lo que no retienen valores entre llamadas

Cuando una variable se crea, se inicializa por defecto al primer valor de su declaración de tipo, en el caso de bit, este valor es '0' y se conservará hasta que se asigne otro

## Lógica combinacional

#### procedimientos

 Un procedimiento (según su codificación) se implementa como un bloque de lógica combinacional o un bloque de lógica secuencial.


```
procedure foo( signal d, gate : in bit, signal q : out bit ) is
  variable z: bit;
begin
  if gate='1' then
    z := d;
  end if;
  q <= z;
end;
...
foo( d, gate, q );
...</pre>
```

```
procedure foo( signal d, gate : in bit, signal q : out bit ) is
begin
  if gate='1' then
    q <= d;
  end if;
end;
...
foo( d, gate, q );
...</pre>
```

- Por ello, se recomienda encapsular 'hardware' mediante componentes.
  - Los subprogramas se reservan para encapsulamiento de funciones 'software'

# Lógica combinacional

#### **ROM**



Lo que no implica que necesariamente se implemente a nivel físico como una ROM,
 ya que esto depende de la tecnología.

```
process (address)
begin
    case address is
    when "00...00" => data <= "...";
    when "00...01" => data <= "...";
    ...
    when others => data <= "...";
    end case;
end process;</pre>
```

#### Multiplexor vectorial 2 a 1

```
library ieee; use ieee.std_logic_1164.all;
entity multiplexer is
  port(
    x0 : in std_logic_vector(7 downto 0);
    x1 : in std_logic_vector(7 downto 0);
    sel : in std_logic;
    y : out std_logic_vector(7 downto 0) );
end multiplexer;


architecture syn1 of multiplexer is
begin
  y <= x1 when sel='1' else x0;
end syn;</pre>
```

```
begin
  process (sel, x0, x1)
  begin
                                architecture syn3 of multiplexer is
    if sel='1' then
                                  signal aux : std logic vector(7 downto 0);
      y \ll x1;
                                begin
    else
                                  aux <= (others=>sel);
      y \le x0;
    end if:
                                  y \le (x1 \text{ and } aux) \text{ or } (x0 \text{ and } not aux);
  end process;
                                end syn3;
end syn2;
```

architecture syn2 of multiplexer is



#### Multiplexor vectorial 2 a 1 genérico





```
library ieee; use ieee.std logic 1164.all;
entity multiplexer is
  generic( n : integer := 8 );
 port(
    x0 : in std_logic_vector(n-1 downto 0);
    x1 : in std_logic_vector(n-1 downto 0);
    sel : in std logic;
        : out std_logic_vector(n-1 downto 0) );
end multiplexer;
architecture syn of multiplexer is
begin
 process (sel, x0, x1)
 begin
    if sel='1' then
      y \le x1;
    else
      y \le x0;
    end if;
  end process;
end syn;
```

#### Multiplexor vectorial 4 a 1 genérico

```
library ieee; use ieee.std logic 1164.all;
entity multiplexer is
  generic( n : integer := 8 );
  port(
    x0, x1, x2, x3 : in std_logic_vector(n-1 downto 0);
                     : in std logic vector(1 downto 0);
    sel
                     : out std logic vector(n-1 downto 0) );
end multiplexer;
architecture syn of multiplexer is
begin
  process (sel, x0, x1, x2, x3)
                                                importante el others para
  begin
                                              cubrir los valores metalógicos
    case sel' is
      when "00" => y \le x0;
      when "01" => y \le x1;
      when "10" => y \le x2;
      when others => y <= x3;
    end case;
  end process;
end syn;
```

#### codificador de prioridad 8 a 3 (i)

```
THE THE PARTY OF T
```

```
library ieee; use ieee.std logic 1164.all;
entity priorityEncoder is
  port(
    x : in std logic vector(7 downto 0);
    y : out std logic vector(2 downto 0);
    gs : out std logic );
                                                    el proceso se activa cuando
end priorityEncoder;
                                                    hay un evento en cualquiera
                                                     de las componentes de x
architecture syn1 of priorityEncoder is
begin
                                        x(7) se chequea en primer lugar
  process (x)
                                              es el más prioritario
  begin
    if
           x(7)='1' then y <= "111"; gs <= '1';
    elsif x(6)='1' then y <= "110"; gs <= '1';
                                                             la anidación de if especifica
    elsif x(5)='1' then y <= "101"; qs <= '1';
                                                             la prioridad de las entradas
    elsif x(4)='1' then y <= "100"; qs <= '1';
    elsif x(3)='1' then y <= "011"; gs <= '1';
    elsif x(2)='1' then y \le "010"; gs \le '1';
    elsif x(1)='1' then y <= "001"; gs <= '1';
    elsif x(0)='1' then y <= "000"; gs <= '1';
    else y <= "000"; gs <= '0';
    end if:
                                                la respuesta cuando no hay entradas
  end process;
                                             activadas se especifica en la última rama else
end syn1;
```

end process;

end syn2;

### **Ejemplos**

#### codificador de prioridad 8 a 3 (ii)

```
library ieee; use ieee.std logic 1164.all;
entity priorityEncoder is
  port(
    x : in std logic vector(7 downto 0);
    y : out std logic vector(2 downto 0);
    gs : out std logic );
end priorityEncoder;
architecture syn2 of priorityEncoder is
begin
  process (x)
  begin
    y <= "000"; qs <= '0';
    if x(0)='1' then y \le "000"; gs \le '1'; end if;
    if x(1)='1' then y <= "001"; gs <= '1'; end if;</pre>
    if x(2)='1' then y \le "010"; qs \le '1'; end if;
    if x(3)='1' then y <= "011"; gs <= '1'; end if;</pre>
    if x(4)='1' then y <= "100"; gs <= '1'; end if;
    if x(5)='1' then y <= "101"; gs <= '1'; end if;
    if x(6)='1' then y <= "110"; gs <= '1'; end if;</pre>
```

if x(7)='1' then y <= "111"; gs <= '1'; end if;

valores 'por defecto' de las salidas cuando no hay entradas activadas

la prioridad de las entradas se especifica sobreescribiendo selectivamente el valor asignado por anteriores sentencias

x(7) se chequea en último lugar es el más prioritario

#### codificador de prioridad genérico

```
THE THE PARTY OF T
```

```
library ieee;
use ieee.std logic 1164.all;
use ieee.numeric_std.all;
entity priorityEncoder is
  generic( n : integer := 3 );
  port(
    x: in std logic vector(2**n-1 downto 0);
    y : out std_logic_vector(n-1 downto 0);
    gs : out std logic );
end priorityEncoder;
architecture syn of priorityEncoder is
begin
                                                    la prioridad de las entradas se especifica
                                                    sobreescribiendo selectivamente el valor
  process (x)
                                                      asignado por anteriores sentencias
  begin
    y <= (others=>'0'); gs <= .!0'
                                                         el rango de i es computable y
    for i in x'reverse range loop <
                                                         el bucle puede desenrrollarse
      if x(i)='1' then
         y <= std logic vector(to unsigned( i, n ));</pre>
         qs <= '1';
                                                              expresión computable
      end if:
    end loop;
  end process;
end syn;
```

# **Ejemplos**

#### decodificador 3 a 8 (i)

```
library ieee; use ieee.std_logic_1164.all;
entity decoder is
  port(
    x : in std_logic_vector(2 downto 0);
    en : in std logic;
    y : out std logic vector(7 downto 0) );
end decoder:
architecture syn1 of decoder is
begin
                                        la selección se anida dentro de la habilitación
  process (x, en)
                                        define un reuso de los cálculos intermedios
  begin
                                           e implica una estructura en cascada
    y <= "00000000";
    if en='1' then
      case x is
        when "000" => y(0) <= '1';
        when "001" => y(1) <= '1';
        when "010" => y(2) <= '1';
        when "011" => y(3) <= '1';
        when "100" => y(4) <= '1';
        when "101" => y(5) <= '1';
        when "110" => y(6) <= '1';
        when others \Rightarrow y(7) \iff '1';
      end case:
    end if;
  end process;
end syn1;
```

# **Ejemplos**

#### decodificador genérico

```
THE THE PARTY OF T
```

```
library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric std.all;
entity decoder is
  generic( n : integer := 3 );
  port(
    x : in std logic vector(n-1 downto 0);
    en : in std logic;
    y : out std logic vector(2**n-1 downto 0) );
end decoder;
architecture syn of decoder is
begin
  process (x, en)
                                            las conversiones de tipo no requieren HW
  begin
                                           ya que sólo indican cómo debe interpretarse
                                                       las señales
    y <= (others=>'0');
    if en='1' then
      y( to integer(unsigned(x)) ) <= '1';</pre>
    end if:
  end process;
end syn;
```

46

# **Ejemplos**

#### decodificador 3 a 8 (ii)



```
library ieee; use ieee.std logic 1164.all;
entity decoder is
  port(
     x : in std logic vector(2 downto 0);
     en : in
                 std_logic;
         : out std logic vector(7 downto 0) );
end decoder;
                                                          cada salida se asigna por separado
architecture syn2 of decoder is
begin
  y(0) \le '1' \text{ when } (en='1' \text{ and } x="000") \text{ else } '0';
  y(1) \le '1' \text{ when } (en='1' \text{ and } x="001") \text{ else } '0';
  y(2) \le '1' \text{ when } (en='1' \text{ and } x="010") \text{ else } '0';
  y(3) \le '1' \text{ when } (en='1' \text{ and } x="011") \text{ else } '0';
  y(4) \le '1' \text{ when } (en='1' \text{ and } x="100") \text{ else } '0';
  y(5) \le '1' \text{ when } (en='1' \text{ and } x="101") \text{ else } '0';
  y(6) \le '1' \text{ when } (en='1' \text{ and } x="110") \text{ else } '0';
  y(7) \le '1' \text{ when } (en='1' \text{ and } x="111") \text{ else } '0';
end syn2;
```

no se comparte la expresión de habilitación luego implica una estructura paralela

# **Ejemplos**

#### decodificador genérico



```
library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;
entity decoder is
  generic( n : integer := 3 );
 port(
    x : in std logic vector(n-1 downto 0);
    en : in std logic;
    y : out std logic vector(2**n-1 downto 0) );
end decoder;
architecture syn of decoder is
begin
  for i in y'range generate
 begin
   y(i) <= '1' when (en='1' and x=unsigned(x)) else '0';
  end generate;
end syn;
```

# **Ejemplos**

#### sumador genérico (i)

```
library ieee;
use ieee.std logic 1164.all;
entity adder is
  generic( n : integer := 8 );
  port(
    x: in std logic vector(n-1 downto 0);
    y: in std logic vector(n-1 downto 0);
    s : out std_logic_vector(n-1 downto 0) );
end adder:
library ieee;
use ieee.numeric std.all;
architecture syn1 of adder is
begin
  s <= std logic vector(unsigned(x) + unsigned(y));</pre>
end syn1;
```

library ieee;
use ieee.std\_logic\_unsigned.all;
architecture syn2 of adder is
begin
 s <= x + y; 4.....
end syn2;</pre>

Según el paquete usado la expresión puede ser más o menos compacta. El HW implementado es el mismo



#### sumador genérico (ii)



```
library ieee;
use ieee.std logic 1164.all;
use ieee.numeric std.all;
entity adder is
  generic( n : integer := 8 );
  port(
         : in std logic vector(n-1 downto 0);
    x
        : in std logic vector(n-1 downto 0);
    cin : in std logic;
         : out std logic vector(n-1 downto 0);
    cout : out std_logic );
end adder;
                                                   para obtener el carry out es necesario
                                                     aumentar en 1 el número de bits
architecture syn of adder is
  signal cinAux : unsigned(0 downto 0);
  signal temp : unsigned(n downto 0);
begin
  cinAux <= (0=>cin);
  temp <= to unsigned( to integer(unsigned(x)) + to integer(unsigned(y))
    + to integer(cinAux), n+1);
  s <= std_logic_vector(temp(n-1 downto 0));</pre>
  cout <= std logic(temp(n));</pre>
end syn;
```

# PE TEN

# Lógica secuencial



- La lógica secuencial puede especificarse mediante:
  - O Procesos sin sentencias wait tales que:
    - Existan señales o variables escritas por el proceso que no se asignen bajo alguna condición de ejecución del mismo.
    - Serán expresiones de flanco o nivel referidas a una señal de reloj las que regulen las condiciones de asignación.
    - Pueden especificar lógica secuencial con temporización (por flanco o nivel) e inicialización (síncrona o asíncrona) de cualquier tipo.
    - Aunque la lista de sensibilidad se ignora, para asegurar la coherencia entre simulación y síntesis, esta debe ser parcial (formada por aquellas señales que disparan cambios de estado: reloj, reset, clear, etc ...).
  - Procesos con sentencias wait
    - Debe tener la forma de wait until referida a una única señal de reloj.
    - Típicamente la sentencia wait debe ser la primera o la última del proceso.
    - Solo puede especificarse lógica secuencial con temporización por flanco y reset síncrono.
    - El standard VHDL define más casos de uso pero no los soporta Xilixn ISE.
  - Asignaciones concurrentes condicionales incompletas (sin rama else)

# Lógica secuencial

#### ejemplos elementales (i)

#### Flip-flop D

```
process
begin
  wait until rising_edge(clk);
  q <= d;
end process;</pre>
```

```
process (clk)
begin
  if rising_edge(clk) then
    q <= d;
  end if;
end process;</pre>
```

#### Flip-flop D con reset asíncrono

```
process (rst, clk)
begin
  if rst='1' then
    q <= '0';
  elsif rising_edge(clk) then
    q <= d;
  end if;
end process;</pre>
```

#### Flip-flop D con reset síncrono

```
process
begin
  wait until rising_edge(clk);
  if rst='1' then
    q <= '0';
  else
    q <= d;
  end if;
end process;</pre>
```

```
process (rst, clk)
begin
  if rising_edge(clk) then
    if rst='1' then
       q <= '0';
    else
       q <= d;
    end if;
end if;
end process;</pre>
```

#### Latch D

```
process (gate, d)
begin
  if gate='1' then
    q <= d;
  end if;
end process;</pre>
```

#### Latch D con reset asíncrono

```
process (rst, gate, d)
begin
  if rst='1' then
    q <= '0';
  elsif gate='1' then
    q <= d;
  end if;
end process;</pre>
```

#### ejemplos elementales (ii)



- Existen expresiones de flanco de subida equivalentes
  - Dispositivos a flanco de bajada tienen expresiones análogas con polaridad invertida.

<pre>process begin   wait until rising_edge(clk);   q &lt;= d; end process;</pre>	<pre>process (clk) begin   if rising_edge(clk) then     q &lt;= d;   end if; end process;</pre>
<pre>process begin   wait until clk'event and clk='1';   q &lt;= d; end process;</pre>	<pre>process (clk) begin   if clk'event and clk='1' then     q &lt;= d;   end if; end process;</pre>
<pre>process begin   wait until not clk'stable and clk='1';   q &lt;= d; end process;</pre>	<pre>process (clk) begin   if not clk'stable and clk='1' then     q &lt;= d;   end if; end process;</pre>
<pre>process begin   wait until clk='1';   q &lt;= d; end process;</pre>	Flip-flop D

#### ejemplos elementales (iii)

- Las asignaciones concurrentes condicionales permiten también especificar lógica secuencial
  - Debido a la equivalencia entre procesos y asignaciones concurrentes

#### Flip-flop D

```
q <= d when rising edge(clk);</pre>
process (clk, d)
begin
  if rising edge(clk) then
    q <= d;
  end if;
end process;
```

#### Flip-flop D con reset asíncrono

```
q <= '0' when rst='1' else</pre>
      d when rising edge(clk);
```

#### Flip-flop D con reset síncrono

```
q <= d when rising edge(clk) else</pre>
    '0' when rst='1';
```

#### Latch D

```
q <= d when gate='1';</pre>
```

```
with clk select
  z <= d when '1',
       z when others;
```

#### Latch D con reset asíncrono

```
q <= '0' when rst='1' else</pre>
      d when gate='1';
```

No obstante, se recomienda el uso de procesos.

#### expresiones de flanco (i)

- Una sentencia if que tenga por condición una especificación de flanco no puede tener rama else.
  - De hecho, en sentencias if-then-elsif la especificación de flanco sólo podrá ser la condición del último if (que no podrá tener rama else).

```
process (clk)
begin
  if rising_edge(clk) then
   q <= ...;
else
   ...
  end if;
end process;</pre>
```

```
process (rst, clk)
begin
  if rst='1' then
    q <= ...;
elsif rising_edge(clk) then
    q <= ...;
else
    ...
end if;
end process;</pre>
```

o En caso contrario la acción especificada debería realizarse en todo momento menos en el preciso instante en el que el reloj cambia, cosa sin sentido en hardware.

#### expresiones de flanco (ii)



 Aunque su presencia en VHDL pueda evaluarse, un evento HW es el punto de división entre dos valores estables diferentes; por tanto, al no ser un valor en sí mismo, no puede utilizarse como argumento en un cálculo.

```
process (rst, clk)
begin
  if rising_edge(clk) or rst='1' then
    q <= ...;
  end if;
end process;</pre>
```

```
process (clk)
begin
  if not rising_edge(clk) then
    q <= ...;
  end if;
end process;</pre>
```

En algunos casos se permite, pero no se recomienda

```
process( rst, clk )
begin
  if rising_edge(clk) and enable='1' then
    q <= ...;
  end if;
end process;</pre>
```

#### expresiones de flanco (ii)

- Una señal puede estar afectada solo por una especificación de flanco.
  - En caso contrario se estaría especificando HW secuencial sensible a varios relojes.


```
process (clk1, clk2)
begin
   if rising_edge(clk1) then
        q <= ...;
   end if;
   if rising_edge(clk2) then
        q <= ...;
   end if;
   end if;
   end process;</pre>
```

 Aún cuando una señal esté afectada por una especificación por flanco, el orden de asignación secuencial de la misma sigue importando.

```
process (rst, clk)
begin
  if rising_edge(clk) then
    q <= d;
  end if;
  if rst='1' then
    q <= '0';
  end if;
end process;</pre>
```

```
d D Q q
```


```
process (rst, clk)
begin
  if rising_edge(clk) then
    q <= d;
  end if;
  q <= not( d );
end process;</pre>
```



# Lógica secuencial FSM (i)




- Todo sistema secuencial puede describirse usando 4 elementos:
  - Un registro de estado.
  - Una red combinacional que calcula el estado siguiente.
  - Una red combinacional que calcula las salidas Moore (dependientes solo del estado).
  - Una red combinacional que calcula las salidas Mealy (dependientes del estado y de la entrada.



# Lógica secuencial FSM (ii)



- Primera alternativa
  - Usar un proceso para especificar cada elemento.



### Lógica secuencial

FSM (iii)

asignar valores por defecto asegura lógica combinacional

```
stateGen:
process (currentState, input)
begin
  nextState <= currentState;</pre>
  case currentState is
    when ... =>
      if (input ...) then
        nextState <= ...;</pre>
      elsif (input ...) then
      else
      end if;
  end case:
end process:
```

```
state:
process (rst n, clk)
begin
  if rst n='0' then
    currentState <= ...;</pre>
  elsif risign edge(clk) then
    currentState <= nextState;</pre>
  end if:
end process;
```

```
mealyGen:
process (currentState, input)
begin
  mealyOutput <= ...;</pre>
  case currentState is
    when ... =>
      if (input ...) then
        mealyOutput <= ...;</pre>
      elsif (input ...) then
      else
      end if:
  end case:
end process:
```


```
mooreGen:
process (currentState)
begin
  mooreOutput <= ...;</pre>
  case currentState is
    when ... =>
      mooreOutput <= ...;</pre>
  end case:
end process;
```

FSM (iv)



#### Segunda alternativa

 Agrupar en un mismo proceso el cálculo del estado siguiente y el cálculo de las salidas tipo Mealy dado que sus estructuras VHDL son equivalentes.



FSM (v)



```
stateGen mealyGen:
process (currentState, input)
begin
  nextState <= currentState;</pre>
  mealyOutput <= ...;</pre>
  case currentState is
    when ... =>
      if (input ...) then
        nextState <= ...;</pre>
        mealyOutput <= ...;</pre>
      elsif (input ...) then
       else
      end if:
  end case:
end process;
```


```
state:
process (rst n, clk)
begin
  if rst n='0' then
    currentState <= ...;</pre>
  elsif risign edge(clk) then
    currentState <= nextState;</pre>
  end if:
end process;
```

```
mooreGen:
process (currentState)
begin
  mooreOutput <= ...;</pre>
  case currentState is
    when ... =>
      mooreOutput <= ...;</pre>
  end case:
end process;
```





Agrupar en un único proceso toda la lógica combinacional.



# **(1)** §

# Lógica secuencial

FSM (vii)



```
stateGen mealyGen mooreGen:
process (currentState, input)
begin
  nextState <= currentState;</pre>
  mealyOutput <= ...;</pre>
  mooreOutput <= ...;</pre>
  case currentState is
    when ... =>
      mooreOutput <= ... v
      if (input ...) then
        nextState <= ...;</pre>
        mealyOutput <= ...; v.
      elsif (input ...) then
      else
      end if:
  end case:
end process;
```

```
state:
process (rst n, clk)
begin
  if rst n='0' then
    currentState <= ...;</pre>
  elsif risign edge(clk) then
    currentState <= nextState;</pre>
  end if:
end process;
```

las salidas tipo Moore sólo dependen del estado

el estado siguiente y las salidas tipo Mealy dependen del estado y de las entradas


# Lógica secuencial

FSM (viii)



#### Cuarta alternativa:

Hacer el cálculo del estado siguiente local al proceso que almacena el estado, dado que el cambio de estado solo se hace efectivo tras eventos del reloj.



### Lógica secuencial

FSM (ix)

```
stateGen state:
process (rst n, clk)
begin
  if rst n='0' then
    currentState <= ...;</pre>
  elsif risign edge(clk) then
    case currentState is
      when ... =>
        if (input ...) then
          currentState <= ...;</pre>
        elsif (input ...) then
        else
        end if;
    end case;
  end if:
end process;
```

ahora no es necesaria la asignación de un valor por defecto a currentState. Si no se asigna conserva su valor

```
mealyGen:
process (currentState, input)
begin
  mealyOutput <= ...;</pre>
  case currentState is
    when ... =>
      if (input ...) then
        mealyOutput <= ...;</pre>
      elsif (input ...) then
      else
      end if:
  end case:
end process;
```


```
mooreGen:
process (currentState)
begin
  mooreOutput <= ...;</pre>
  case currentState is
    when ... =>
      mooreOutput <= ...;</pre>
  end case;
end process:
```

# Lógica secuencial FSM (x)



#### Quinta alternativa:

 Agrupar el proceso que almacena el estado con el que calcula las salidas tipo Moore.



FSM (xi)

```
mooreGen_state:
process (rst_n, clk, currentState)
begin
  mooreOutput <= ...;
  case currentState is
    when ... =>
     mooreOutput <= ...
end case;
if rst_n='0' then
    currentState <= ...;
elsif rising_edge(clk) then
    currentState <= nextState;
end if;
end process;</pre>
```

```
mealyGen:
process (currentState, input)
begin
  mealyOutput <= ...;
  case currentState is
   when ... =>
      if (input ...) then
        mealyOutput <= ...;
   elsif (input ...) then
   else
   end if;
end case;
end process;</pre>
```

```
stateGen:
process (currentState, input)
begin
  nextState <= currentState;
  case currentState is
    when ... =>
        if (input ...) then
            nextState <= ...;
    elsif (input ...) then
        else
        end if;
  end case;
end process;</pre>
```


necesaria para que simule correctamente

el cálculo de la salida se realiza incondicionalmente, si se anidara dentro del if, se interpretaría que dichas salidas hay también que almacenarlas (aunque sólo cambien tras el cambio de estado)

FSM (xii)

Sexta alternativa





FSM (xiii)

```
stateGen MealyGen state:
process (rst n, clk, currentState, input)
begin
  mealyOutput <= ...;</pre>
  case currentState is
    when ... =>
      if (input ...) then
        mealyOutput <= ...;</pre>
      elsif (input ...) then
      else
      end if:
  end case;
  if rst n='0' then
    currentState <= ...;</pre>
  elsif risign_edge(clk) then
    case currentState is
      when ... =>
        if (input ...) then
           currentState <= ...;</pre>
        elsif (input ...) then
        else
        end if:
    end case:
  end if:
end process:
```

```
mooreGen:
process (currentState)
begin
  mooreOutput <= ...;
  case currentState is
    when ... =>
    mooreOutput <= ...;
  end case;
end process;</pre>
```

necesaria para que simule correctamente


necesaria para que simule correctamente

el cálculo de la salida se realiza incondicionalmente

FSM (xiv)



- Séptima alternativa:
  - o Especificar todo usando un único proceso



# Lógica secuencial

FSM (xv)

```
THE STATE OF THE S
```

```
stateGen mealyGen mooreState state:
process (rst_n, clk, currentState, input )
begin
  mealyOutput <= ...;</pre>
  mooreOutput <= ...;
  case currentState is
    when ... =>
      mooreOutput <= ...;</pre>
      if (input ...) then
        mealyOutput <= ...;</pre>
      elsif (input ...) then
      else
      end if:
  end case:
  if rst n='0' then
    currentState <= ...;</pre>
  elsif risign edge(clk) then
    case currentState is
      when ... =>
        if (input ...) then
          currentState <= ...;</pre>
        elsif (input ...) then
        else
        end if:
    end case:
  end if:
end process;
```

el cálculo de la salida se realiza incondicionalmente

Aunque resulte poco intuitivo, este es el método fiable de especificación de HW a nivel RT más abstracto, existen otros pero son difíciles de controlar

# Lógica secuencial

FSM (xvi)



```
stateGen_mealyGen_mooreState_state:
process (rst n, clk input )
  variable currentState :
begin
  mealyOutput <= ...;</pre>
  mooreOutput <= ...;</pre>
  case currentState is
    when ... =>
      mooreOutput <= ...;</pre>
      if (input ...) then
        mealyOutput <= ...;</pre>
      elsif (input ...) then
      else
      end if:
  end case;
  if rst n='0' then
    currentState := ...;
  elsif risign_edge(clk) then
    case currentState is
      when ... =>
        if (input ...) then
          currentState := ...;
        elsif (input ...) then
        else
        end if:
    end case:
  end if:
end process:
```

Adicionalmente, dado que el estado actual es local al proceso, esta señal se puede sustituir por variable

En este caso se estrictamente necesario respetar el orden de los bloques, de manera que la asignación de currentState sea posterior a su lectura

# Lógica secuencial RAM (i)



- Es común especificar lógica secuencial en forma de RAM
  - Lo que no implica que necesariamente se implemente a nivel físico como una RAM,
     ya que esto depende de la tecnología.
  - De hecho, cuando la tecnología objetivo dispone de RAM físicas, lo más práctico es instanciarlas como componentes.

RAM con lectura y escritura asíncrona (puertos de datos separados)

```
architecture ...;
...
  type ramType is array (0 to ...) of std_logic_vector(... downto 0);
  signal ram : ramType;
begin
  ...
  process (we, dataIn, address)
  begin
    if we='1' then
      ram( to_integer( unsigned( address ) ) ) <= dataIn;
    end if;
end process;
dataOut <= ram( to_integer( unsigned( address ) ) );
...
end;</pre>
```

### Lógica secuencial

### RAM (ii)



RAM con lectura y escritura asíncrona (puerto de datos único)

```
process (we, data)
begin
  if we='1' then
    ram( to_integer( unsigned( address ) ) ) <= data;
  end if;
end process;
data <= ram( to_integer( unsigned( address ) ) ) when re='1' else (others => 'Z');
```

RAM con lectura asíncrona y escritura síncrona (puerto de datos separado)

```
process (clk)
begin
  if rising_edge(clk) then
    if we='1' then
      ram( to_integer( unsigned( address ) ) ) <= dataIn;
    end if;
  end if;
end process;
dataOut <= ram( to_integer( unsigned( address ) ) );</pre>
```

RAM con lectura y escritura síncrona (puerto de datos separado)

```
process (clk)
begin
   if rising_edge(clk) then
      if we='1' then
        ram( to_integer( unsigned( address ) ) ) <= dataIn;
   end if;
   dataOut <= ram( to_integer( unsigned( address ) ) );
   end if;
end process;</pre>
```

#### registro genérico

```
library ieee; use ieee.std_logic_1164.all;
entity reg is
 generic( n : integer := 8 );
 port(
   rst_n, clk, ld : in std_logic;
```

```
architecture syn2 of reg is
begin
  process (rst n, clk)
  begin
    if rst n='0' then
      dout <= (others=>'0');
    elsif rising edge(clk) then
      if ld='1' then
        dout <= din;
      end if:
    end if:
  end process;
end syn2;
```

# **Ejemplos**

#### registro genérico con salida en alta impedancia

```
library ieee; use ieee.std_logic_1164.all;
entity triStateReg is
  generic( n : integer := 8 );
 port(
    rst n, clk, ld, en : in std logic;
                        : in std logic vector( n-1 downto 0 );
    din
    dout
                         out std_logic_vector( n-1 downto 0 ) );
end triStateReq;
architecture syn1 of triStateReg is
  signal cs, ns : std logic vector(n-1 downto 0);
begin
 process (rst n, clk)
 begin
    if rst_n='0' then
      cs <= (others=>'0');
    elsif rising edge(clk) then
      cs <= ns;
    end if:
 end process;
 ns <= din when ld='1'
            else cs;
 dout <= cs when en='1'</pre>
          else (others=>'Z');
end syn1;
```

la señal cs se necesita para distinguir el estado que puede inicializarse y cargarse de modo independiente a la salida que puede desabilitarse

```
architecture syn2 of triStateReg is
  signal cs : std_logic_vector(n-1 downto 0);
begin
  process (rst_n, clk, en, cs)
  begin
    dout <= (others=>'Z');
    if en='1' then
      dout <= cs;
    end if:
    if rst n='0' then
      cs <= (others=>'0');
    elsif rising_edge(clk) then
      if ld='1' then
        cs <= din;
      end if:
    end if:
  end process;
end syn2;
```

### **Ejemplos**


#### registro de desplazamiento genérico

library ieee; use ieee.std logic 1164.all;



```
entity shiftReg is
  generic( n : integer := 8 );
  port(
    rst_n, clk, sht : in std_logic;
                    : in std logic;
    din
                    : out std_logic_vector(n-1 downto 0) );
    dout
end shiftReq;
architecture syn1 of shiftReg is
  signal cs, ns : std logic vector(n-1 downto 0);
begin
  process (rst_n, clk)
                                           architecture syn2 of shiftReg is
  begin
                                            signal cs : std logic vector(n-1 downto 0);
    if rst n='0' then
                                          begin
      cs <= (others=>'0');
                                            process (rst n, clk)
    elsif rising edge(clk) then
                                            begin
      cs <= ns;
                                               if rst n='0' then
    end if;
  end process:
                                                 cs <= (others=>'0');
  process (cs, sht, din)
                                               elsif rising edge(clk) then
  begin
                                                 if sht='1' then
    ns <= cs;
                                                   for i in cs'high downto cs'low+1
    if sht='1' then
                                                   loop
      for i in ns'high downto ns'low+1
                                                     cs(i) \le cs(i-1);
      loop
                                                   end loop:
        ns(i) \le cs(i-1);
                                                  cs(0) <= din;
      end loop:
      ns(0) <= din;
                                                 end if:
    end if:
                                               end if:
  end process:
                                            end process;
  dout <= cs;
                                            dout <= cs;
end syn1;
                                          end syn2;
```

#### contador ascendente genérico (i)





### **Ejemplos**

### contador ascendente genérico (ii)



```
architecture syn2 of counter is
  signal cs : unsigned(n-1 downto 0); ... signal cs : integer;
begin
  process (rst_n, clk, ce, cs)
  begin
   if ce='1' and cs=2**n-1 then
      tc <= '1';
    else
    tc <= '0';
   end if:
   if rst_n='0' then
      cs <= (others=>'0'); .....
    elsif rising edge(clk) then
      if ld='1' then
        cs <= unsigned(din);.....</pre>
      elsif ce='1' then
      cs <= cs + 1;
      end if:
    end if:
  end process;
end syn2;
```

```
architecture syn3 of counter is
                                     begin
                                       process (rst_n, clk, ce, cs)
                                       begin
dout <= std_logic_vector(cs); dout <= std_logic_vector(to_unsigned(cs,n));</pre>
                                         if ce='1' and cs=2**n-1 then
                                           tc <= '1';
                                         else
                                           tc <= '0';
                                         end if:
                                         if rst_n='0' then
                                     ..... cs <= 0;
                                         elsif rising_edge(clk) then
                                            if ld='1' then
                                       .....> cs <= to_integer(unsigned(din));</pre>
                                           elsif ce='1' then
                                             cs <= cs + 1;
                                           end if:
                                         end if:
                                       end process;
                                     end syn3;
```

begin

begin

process (rst\_n, clk, ce, cs)

cs <= (others=>'0');

if cs=max then

cs <= cs + 1;

elsif rising edge(clk) then

cs <= (others=>'0');

if rst n='0' then

if ce='1' then

else

end if:

end if:

end syn1;

end process;

end if:

dout <= std\_logic\_vector(cs);</pre>

### **Ejemplos**

#### contador ascendente modulo-máximo genérico


```
library ieee; use ieee.std_logic_1164.all; use ieee.numeric_std.all;
entity counter is
  generic(
    n : integer := 4; max : integer := 10 );
  port(
    rst_n, clk, ce : in std_logic;
    dout : out std_logic_vector(n-1 downto 0));
end counter;

architecture syn1 of counter is
  signal cs : unsigned(n-1 downto 0);
```


```
architecture syn2 of counter is
   signal cs : unsigned(n-1 downto 0);
begin
   process (rst_n, clk, ce, cs)
   begin
    dout <= std_logic_vector(cs);
   if rst_n='0' then
       cs <= (others=>'0');
   elsif rising_edge(clk) then
       if ce='1' then
       cs <= (cs + 1) mod max;
   end if;
   end process;
end syn2;</pre>
```


# **Ejemplos**FSM temporizadas

- Se desea diseñar un sistema digital que controle un semáforo con botón peatonal para solicitar el paso:
  - El semáforo de coches estará verde como mínimo un periodo verde y continuará en verde hasta que no se pulse el botón.
  - Si se pulsa el botón, sólo si el semáforo de coches ha estado en verde durante un periodo verde completo, el semáforo de coches pasará a amarillo durante un periodo amarillo, tras el cual se pondrá en rojo. Entonces el semáforo de peatones pasará a verde.
  - El semáforo de coches permanecerá en rojo un periodo rojo, tras el cual el semáforo de peatones pasará a rojo. El semáforo de coches pasará a verde transcurrido un periodo amarillo desde que el de peatones cambió.



### FSM temporizadas: controlador





#### FSM temporizadas: codificación con 7 sentencias



```
library ieee;
use ieee.std logic 1164.all;
use ieee.numeric_std.all;
entity controlSemaforo is
  generic( cVerde, cAmarillo, cVerde : natural );
 port(
   clk, rst : in std logic;
   boton: in std logic;
   coche, peaton : out std logic vector(2 downto 0)
end controlSemaforo;
architecture syn of controlSemaforo is
 constant sRojo
                     : std logic vector(2 downto 0) := "100";
 constant sAmarillo : std logic vector(2 downto 0) := "010";
                     : std logic vector(2 downto 0) := "001";
  constant sVerde
 signal cargar, fin : std_logic;
 signal csT, nsT, ciclos : natural;
 type estados_t is ( cVpR, cVpRCond, cApR, cRpV, cRpR );
 signal csC, nsC : estados_t;
begin
end syn;
```

### FSM temporizadas: temporizador VHDL



```
stateTemporizador:
process (rst, clk)
begin
  if rst = '1' then
    csT <= cVerde;
  elsif rising_edge(clk) then
    csT <= nsT;
  end if;
end process;</pre>
```

```
mooreGenTemporizador:
process (csT)
begin
  if csT=0 then
    fin <= '1';
  else
    fin <= '0';
  end if;
end process;</pre>
```

```
stateGenTemporizador:
process (csT, fin, cargar, ciclos)
begin
  if cargar='1' then
    nsT <= ciclos;
elsif fin='1' then
    nsT <= csT;
else
    nsT <= csT - 1;
end if;
end process;</pre>
```

### FSM temporizadas: controlador VHDL (i)

```
stateControlador:
process (rst, clk)
begin
  if rst='1' then
    csC <= cVpR;
  elsif rising_edge(clk) then
    csC <= nsC;
  end if;
end process;</pre>
```

```
stateGenControlador:
process (csC, boton, fin)
begin
  nsC <= csC;</pre>
  case csC is
    when cVpR =>
      if fin='1' then
        nsC <= cVpRCond;</pre>
      end if;
    when cVpRCond =>
      if boton='1' then
        nsC <= cApR;</pre>
      end if;
    when cApR =>
      if fin='1' then
        nsC <= cRpV;</pre>
      end if;
    when cRpV =>
      if fin='1' then
        nsC <= cRpR;</pre>
      end if;
    when cRpR =>
      if fin = '1' then
        nsC <= cVpR;</pre>
      end if:
  end case:
end process;
```





#### FSM temporizadas: controlador VHDL (ii)

```
mooreGenControlador:
process (csC)
begin
  case csC is
    when cVpR =>
       coche <= sVerde;</pre>
       peaton <= sRojo;</pre>
    when cVpRCond =>
       coche <= sVerde;</pre>
       peaton <= sRojo;</pre>
    when cApR =>
       coche <= sAmarillo;</pre>
       peaton <= sRojo;</pre>
    when cRpV =>
       coche <= sRojo;</pre>
       peaton <= sVerde;</pre>
    when cRpR =>
       coche <= sRojo;</pre>
       peaton <= sRojo;</pre>
  end case:
end process;
```

```
mealyGenControlador:
process (csC, boton, fin)
begin
  cargar <= '0';</pre>
  ciclos <= cVerde;
  case csC is
    when cVpR =>
      null:
    when cVpRCond =>
      if boton='1' then
        cargar <= '1';
        ciclos <= cAmarillo;</pre>
      end if;
    when cApR =>
      if fin='1' then
        cargar <= '1';
        ciclos <= cRojo;
      end if:
    when cRpV =>
      if fin='1' then
        cargar <= '1';</pre>
        ciclos <= cAmarillo;</pre>
      end if:
    when cRpR =>
      if fin='1' then
        cargar <= '1';</pre>
        ciclos <= cVerde;
      end if:
  end case:
end process;
```



### FSM temporizadas: codificación con 2 sentencias



```
library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric std.all;
entity controlSemaforo is
 generic( cVerde, cAmarillo, cVerde : natural );
  port(
    clk, rst : in std logic;
   boton : in std_logic;
    coche, peaton : out std_logic_vector(2 downto 0)
  );
end controlSemaforo;
architecture syn of controlSemaforo is
  constant sRojo
                     : std logic vector(2 downto 0) := "100";
  constant sAmarillo : std logic vector(2 downto 0) := "010";
                     : std logic vector(2 downto 0) := "001";
  constant sVerde
  signal cargar, fin : std logic;
  signal numCiclos, ciclos : natural;
  type estados_t is ( pVsR, pVsRCond, pAsR, pRsV, pRsA );
  signal estado : estados t;
begin
end syn;
```



# **Ejemplos**

### FSM temporizadas: temporizador VHDL unificado



```
temporizador:
process (rst, clk)
begin
  if numCiclos=0 then
    fin <= '1';
  else
    fin <= '0';
  end if;
  if rst='1' then
    numCiclos <= cVerde;</pre>
  elsif rising edge(clk) then
    if cargar='1' then
      numCiclos <= ciclos;</pre>
    elsif fin='0' then
      numCiclos <= numCiclos - 1;</pre>
    end if:
  end if:
end process;
```

### FSM temporizadas: controlador VHDL unificado


```
controlador:
process (rst, clk, estado, boton, fin)
begin
  cargar <= '0'; ciclos <= cVerde;</pre>
  case estado is
    when cVpR =>
      coche <= sVerde; peaton <= sRojo;</pre>
    when cVpRCond =>
      coche <= sVerde; peaton <= sRojo;</pre>
       if boton='1' then
         cargar <= '1'; ciclos <= cAmarillo;</pre>
      end if:
    when cApR =>
      coche <= sAmarillo; peaton <= sRojo;</pre>
      if fin='1' then
        cargar <= '1'; ciclos <= cRojo';</pre>
      end if:
    when cRpV =>
      coche <= sRojo; peaton <= sVerde;</pre>
       if fin='1' then
         cargar <= '1'; ciclos <= cAmarillo;</pre>
      end if:
    when cRpR =>
      coche <= sRojo; peaton <= sRojo;</pre>
       if fin='1' then
         cargar <= '1'; ciclos <= cVerde;</pre>
      end if:
  end case:
```

```
if rst = '1' then
    estado <= cVpR;</pre>
  elsif rising edge(clk) then
    case estado is
      when cVpR =>
        if fin='1' then
          estado <= cVpRCond;</pre>
        end if:
      when cVpRCond =>
        if boton='1' then
          estado <= cApR;
        end if:
      when cApR =>
        if fin='1' then
          estado <= cRpV;
        end if:
      when cRpV =>
        if fin='1' then
          estado <= cRpR;
        end if:
      when cRpR =>
        if fin='1' then
          estado <= cVpR;
        end if:
    end case:
  end if:
end process;
```

end syn;

### **Ejemplos**

#### FSMT: codificación con 1 sentencia





#### FSMT: temporizador + controlador unificados (i)

```
fsmt:
process (rst, clk, estado)
begin
  case estado is
    when cVpR =>
       coche <= sVerde;</pre>
       peaton <= sRojo;</pre>
    when cVpRCond =>
       coche <= sVerde;</pre>
       peaton <= sRojo;</pre>
    when cApR =>
       coche <= sAmarillo;</pre>
       peaton <= sRojo;</pre>
    when cRpV =>
       coche <= sRojo;
       peaton <= sVerde;</pre>
    when pRpR =>
       coche <= sRojo;</pre>
       peaton <= sRojo;</pre>
  end case:
```

```
if rst = '1' then
               <= cVpR;
    estado
    numCiclos <= cVerde;</pre>
  elsif rising_edge(clk) then
    if numCiclos /= 0 then
      numCiclos <= numCiclos - 1;</pre>
    else
      case estado is
        when cVpR =>
           estado <= cVpRCond;</pre>
        when cVpRCond =>
           if boton='1' then
                        <= cApR;
             estado
             numCiclos <= cAmarillo;</pre>
           end if:
        when cApR =>
           estado
                      <= cRpV;
          numCiclos <= cRojo;</pre>
        when cRpV =>
           estado
                      <= cRpR;
          numCiclos <= cAmarillo;</pre>
        when cRpR =>
           estado
                      <= cVpR;
          numCiclos <= cVerde;</pre>
      end case:
    end if:
  end if;
end process;
```



# **Ejemplos**

#### FSMT: temporizador + controlador unificados (ii)

```
fsmt:
process (rst, clk, estado)
begin
if rst = '1' then
    estado
                <= cVpR;
    numCiclos <= cVerde;</pre>
  elsif rising edge(clk) then
    if numCiclos /= 0 then
      numCiclos <= numCiclos - 1;</pre>
    else
       case estado is
         when cVpR =>
           coche <= sVerde;</pre>
           peaton <= sRojo;</pre>
           estado <= cVpRCond;</pre>
         when cVpRCond =>
           coche <= sVerde;</pre>
           peaton <= sRojo;</pre>
           if boton='1' then
              estado
                         <= cApR;
             numCiclos <= cAmarillo;</pre>
           end if:
```

```
when cApR =>
                    <= sAmarillo;
          coche
                    <= sRojo;
          peaton
          estado
                    <= cRpV;
          numCiclos <= cRojo;</pre>
        when cRpV =>
                    <= sRojo;
          coche
                    <= sVerde;
          peaton
          estado
                    <= cRpR;
          numCiclos <= cAmarillo;</pre>
        when cRpR =>
                    <= sRojo;
          coche
                    <= sRojo;
          peaton
                    <= cVpR;
          estado
          numCiclos <= cVerde;</pre>
      end case:
    end if:
 end if:
end process;
                 INCORRECTO
```

botón coche / 3 / 3 controlador fin temporizador

### FSMT: codificación con 1 sentencia y variables



```
library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;
entity controlSemaforo is
 generic( cVerde, cAmarillo, cVerde : natural );
 port(
    clk, rst : in std_logic;
   boton : in std logic;
    coche, peaton : out std_logic_vector(2 downto 0)
  );
end controlSemaforo;
architecture syn of controlSemaforo is
                     : std logic vector(2 downto 0) := "100";
  constant sRojo
  constant sAmarillo : std logic vector(2 downto 0) := "010";
                     : std_logic_vector(2 downto 0) := "001";
  constant sVerde
begin
end syn;
```

### **Ejemplos**

#### FSMT: temporizador + controlador unificados y variables


```
fsmt:
process (rst, clk)
  type estados_t is
   ( cVpR, cVpRCond, cApR,cRpV, cRpR );
  variable estado : estados t;
  variable numCiclos : natural;
begin
  case estado is
    when cVpR =>
       coche <= sVerde;</pre>
      peaton <= sRojo;</pre>
    when cVpRCond =>
       coche <= sVerde;
      peaton <= sRojo;</pre>
    when cApR =>
       coche <= sAmarillo;</pre>
      peaton <= sRojo;</pre>
    when cRpV =>
      coche <= sRojo;</pre>
      peaton <= sVerde;</pre>
    when cRpR =>
      coche <= sRojo;
      peaton <= sRojo;</pre>
  end case:
```


```
if rst = '1' then
    estado
              := cVpR;
   numCiclos := cVerde;
  elsif rising_edge(clk) then
    if numCiclos /= 0 then
      numCiclos := numCiclos - 1:
    else
      case estado is
        when cVpR =>
          estado := cVpRCond;
        when cVpRCond =>
          if boton='1' then
                      := cApR;
            estado
            numCiclos := cAmarillo;
          end if:
        when cApR =>
                    := cRpV;
          estado
          numCiclos := cRojo;
        when cRpV =>
                    := cRpR;
          estado
          numCiclos := cAmarillo;
        when cRpR =>
          estado
                    := cVpR;
          numCiclos := cVerde;
      end case:
    end if:
  end if:
end process:
```

# **Ejemplos**

#### FSMD: FSM + ruta de datos

 Se desea diseñar un multiplicador sin signo por el algoritmo de suma y desplazamiento.






(A)

#### DAS

# **Ejemplos**


FSMD: estructura





# **Ejemplos**

#### FSMD: controlador



```
begin
 A := Ain;
  B := Bin;
  R := 0;
 for C:=0 to 31 do begin
    if B_0=1 then R := R+A;
     A := A << 1;
  B := B >> 1;
  end for;
  Rout := R;
end;
```

#### FSMD: codificación con datapath explícito (i)

```
library ieee;
use ieee.std logic 1164.all;
use ieee.numeric std.all;
entity multiplier is
 port
            in std logic;
   rst n:
   clk
            in std logic;
             in std logic;
    start:
   done : out std logic;
             in std logic vector(31 downto 0);
             in std logic vector(31 downto 0);
    b
          : out std_logic_vector(63 downto 0)
  );
end multiplier;
architecture syn of multiplier is
  signal ra, ra next : unsigned(63 downto 0);
  signal rb, rb next : unsigned(31 downto 0);
  signal rr, rb next : unsigned(63 downto 0);
  signal rc, rc_next : unsigned(4 downto 0);
 type states t is ( s0, s1, s2, s3, s4 );
  signal state : states_t; ...
begin
end syn;
```

TO THE PARTY OF TH

estado actual y siguiente de los registros de la ruta de datos

estado actual del controlador

### FSMD: codificación con datapath explícito (ii)



#### Controlador

```
process (rst_n, clk)
begin
  if rst n='0' then
    state <= s0;
 elsif rising edge(clk) then
    case state is
      when s0 =>
        if start='1' then
          state <= s1;
        end if:
      when s1 =>
        if rb(0)='1' then
          state <= s3;</pre>
        else
          state <= s2;
        end if:
      when s2 | s3 =>
        state <= s4;
```

```
when s4 =>
   if rc=0 then
      state <= s0;
   else
      if rb(0)='1' then
        state <= s3;
      else
        state <= s2;
      end if;
   end case;
   end process;</pre>
```

#### FSMD: codificación con datapath explícito (iii)



#### Ruta de datos (registros)

```
process (rst_n, clk)
begin
  if rst_n='0' then
    ra <= (others => '0');
    rb <= (others => '0');
    rr <= (others => '0');
    rc <= (others => '0');
    elsif rising_edge(clk) then
      ra <= ra_next;
    rb <= rb_next;
    rr <= rr_next;
    rc <= rc_next;
end if;
end process;</pre>
```

el reuso de recursos es explícito y no depende de la herramienta EDA

#### Ruta de datos (elementos combinacionales)

```
done <= '1' when state=s0 else '0';</pre>
r <= std_logic_vector(rr);
with state select
  ra next <=
    resize(unsigned(a), 64) when s0,
    ra(62 downto 0) & '0' when s2 | s3,
    ra when others;
with state select
  rb next <=
    unsigned(b) when s0,
    '0' & rb(31 downto 1) when s2 | s3,
    rb when others;
with state select
  rr next <=
    (others => '0') when s1,
    ra + rr when s3,
    rr when others;
with state select
  rc next <=
 rc + 1 when s2 | s3,
    rc when others;
```

#### FSMD: codificación con datapath implícito (i)

```
library ieee;
use ieee.std logic 1164.all;
use ieee.numeric std.all;
entity multiplier is
  port
            in std logic;
    rst n:
    clk
            in std logic;
              in std logic;
    start:
    done : out std logic;
             in std logic vector(31 downto 0);
              in std logic vector(31 downto 0);
    b
          : out std_logic_vector(63 downto 0)
end multiplier;
architecture syn of multiplier is
  signal ra : unsigned(63 downto 0);
  signal rb : unsigned(31 downto 0);
                                               estado actual y de los
  signal rr : unsigned(63 downto 0);
                                            registros de la ruta de datos
  signal rc : unsigned(4 downto 0);
  type states t is ( s0, s1, s2, s3, s4 );
  signal state : states_t; ...
begin
                                                     estado actual del controlador
end syn;
```

### **Ejemplos**

#### FSMD: codificación con datapath implícito (ii)



```
done <= '1' when state=s0 else '0';</pre>
r <= std_logic_vector(rr);
process (rst_n, clk, cs)
begin
  if rst n='0' then
    ra <= (others => '0');
    rb <= (others => '0');
    rr <= (others => '0');
    rc <= (others => '0');
    state <= s0;
  elsif rising edge(clk) then
    case state is
      when s0 =>
        ra <= resize( unsigned(a), 64);</pre>
        rb <= unsigned(b);</pre>
        if start='1' then
          state <= s1;
        end if:
      when s1 =>
        rr <= (others => '0');
        if rb(0)='1' then
           state <= s3;.....
                         el reuso de recursos es implícito y
        else
           state <= s2; depende de la herramienta EDA
        end if:
```

```
when s2 =>
        ra <= ra(62 downto 0) & '0';
        rb <= '0' & rb(31 downto 1);
        rc <= rc + 1;
        state <= s4;
      when s3 =>
        rr <= ra + rr;
        ra <= ra(62 downto 0) & '0';
        rb <= '0' & rb(31 downto 1);
        rc <= rc + 1;
        state <= s4;
      when s4 =>
        if rc=0 then
          state <= s0;
        else
          if rb(0)='1' then
            state <= s3;
          else
            state <= s2;
          end if:
        end if:
   end case;
 end if;
end process;
```

#### FSMD: codificación VHDL con datapath implícito y variables

```
process (rst_n, clk)
  variable ra : unsigned(63 downto 0);
  variable rb : unsigned(31 downto 0);
  variable rr : unsigned(63 downto 0);
  variable rc : unsigned(4 downto 0);
  type states_t is ( s0, s1, s2, s3, s4 );
  variable state : states_t;
begin
  r <= std_logic_vector(rr);
  if state=s0 then
    done <= '1';
  else
    done <= '1';
  end if;
  if rst n='0' then
    ra := (others => '0');
    rb := (others => '0');
    rr := (others => '0');
    rc := (others => '0');
    state := s0;
  elsif rising edge(clk) then
    case state is
      when s0 =>
        ra := resize( unsigned(a), 64);
        rb := unsigned(b);
        if start='1' then
          state := s1;
        end if;
```

```
when s1 =>
        rr := (others => '0');
        if rb(0)='1' then
          state := s3;
        else
          state := s2;
        end if:
      when s2 =>
        ra := ra(62 downto 0) & '0';
        rb := '0' \& rb(31 downto 1);
        rc := rc + 1;
        state := s4;
      when s3 =>
        rr := ra + rr;
        ra := ra(62 downto 0) & '0';
        rb := '0' \& rb(31 downto 1);
        rc := rc + 1;
        state := s4;
      when s4 =>
        if rc=0 then
          state := s0;
        else
          if rb(0)='1' then
            state := s3;
          else
            state := s2;
          end if:
        end if:
    end case:
  end if:
end process;
```

# tema 4:

#### DAS

### **Mezclando VHDL**

#### directivas: atributos y pragmas

- La mayor parte de las herramientas EDA pueden ser parcialmente controladas desde el propio código VHDL
  - Caracterización del entorno de funcionamiento del circuito.
  - Ligaduras u opciones del proceso de síntesis.
  - o Control sobre el modo en que se interpretan las construcciones VHDL.
- Habitualmente existe 2 maneras:
  - Mediante atributos VHDL

```
attribute OPT_MODE : string;
attribute OPT_MODE of cronometro : entity is "area"
```

 Mediante pragmas, comentarios VHDL que tienen un significado especial para la herramienta de síntesis

```
-- pragma translate_off
-- pragma translate_on
```

### Mezclando VHDL

#### componentes prediseñados (i)



- No todo el diseño debe ser especificado en VHDL:
  - Las herramientas EDA permiten mezclar código con otros mecanismos de especificación
    - Esquemáticos, diagramas de estados, módulos prediseñados, uso de otros lenguajes
  - Desde el punto de vista VHDL:
    - Estos comportamientos se encapsulan en bibliotecas
    - Se instancian como componentes

#### Ventajas:

 Se puede ahorrar tiempo y esfuerzo (discutible, excepto en caso de módulos prediseñados)

#### **Problemas:**

- Pérdida de portabilidad: VHDL "simulable" es estándar, VHDL "sintetizable" más o menos, los restantes mecanismos de especificación son dependientes de herramienta.
- Dependencia tecnológica: muchos módulos prediseñados pueden sólo ser aplicables para ciertas tecnologías objetivo.
- Necesidad de co-simulación: se necesitan la interacción de varios simuladores cada uno especializado en una representación
  - Existen generadores de modelos simulables VHDL.

#### ىتى DAS



#### componentes prediseñados (ii)



- Existen diferentes tipos de componentes prediseñados:
  - Soft-macros: especificaciones (esquemáticos o descripciones HDL) que se mezclan y sintetizan con el resto de los componentes del sistema.
    - No se puede garantizar su rendimiento.
  - Hard-macros: bloques presintetizados (típicamente netlist) que incluyen datos relativos a emplazamiento y rutado.
    - Su rendimiento puede garantizarse.
  - Hardwired-macros: bloques prefabricados y predifundidos sobre silicio
    - Su rendimiento está completamente caracterizado.
- Los componentes prediseñados pueden
  - Estar almacenados en bibliotecas de módulos o crearse por un generador de módulos.
  - Tener pinout, funcionalidad y rendimiento fijo o parametrizable.
  - La parametrización puede ser realizada explícitamente por el diseñador o implícitamente por la herramientas EDA según las ligaduras de diseño.
  - Su uso puede ser gratuito o puede requerir el pago de licencias (según el uso).
  - Pueden tener diversos grados de complejidad:
    - Primitivos: celdas elementales (AND, FF, ...) proyectables directamente sobre el HW.
    - IP-cores: bloques de alta complejidad diseñados por compañías independientes.

### **Mezclando VHDL**

#### componentes prediseñados (iii)

- Los componentes prediseñados pueden usarse y parametrizarse por:
  - Instanciación directa.
  - Inferencia a partir de un operador, una función o un fragmento de código que responda a una cierta estructura.

```
architecture ...:
  component MULT18X18
  port ( a : in std logic vector(17 downto 0);
         b: in std logic vector(17 downto 0);
         p : out std logic vector(35 downto 0));
  end component;
begin
  multiplier: MULT18X18
    port map (a => leftOp, b => rightOp, p => product);
end ...;
                                            en ambos casos se usará uno de los
architecture ...:
                                            multiplicadores 18x18 predifundidos
                                              de los que dispone la Spartan 3
 attribute MULT STYLE : string;
 attribute MULT STYLE of product : signal is "block";
begin
  product <= leftOp * rigthOp;</pre>
end ...;
```

108

### Acerca de Creative Commons





- Ofrece algunos derechos a terceras personas bajo ciertas condiciones. Este documento tiene establecidas las siguientes:
  - Reconocimiento (Attribution):
    En cualquier explotación de la obra autorizada por la licencia hará falta reconocer la autoría.
  - No comercial (Non commercial):

    La explotación de la obra queda limitada a usos no comerciales.
  - Compartir igual (Share alike):

    La explotación autorizada incluye la creación de obras derivadas siempre que mantengan la misma licencia al ser divulgadas.

Más información: https://creativecommons.org/licenses/by-nc-sa/4.0/